ExcelTabellen automatisch aus Accesss aktualisieren  Teil 1
Versionen: Access 2007, 2003, 2002/XP und 2000

Wenn Sie Access-Tabellen, CSV-Dateien oder Outlook-Ordner über die Funktion "Externe Daten", "Tabellen verknüpfen" in eine Datenbank einbinden, können Sie Inhalte der verknüpften Tabellen beliebig bearbeiten und Datensätze ändern, löschen oder hinzufügen. Access kümmert sich dann automatisch im Hintergrund darum, dass die Daten in den zugrunde liegenden Tabellen bzw. Dateien aktualisiert werden. Das war zunächst auch mit Excel-Arbeitsmappen möglich, aber 2005 verlor Microsoft einen Patentstreit um diese Technik und musste die Funktionalität daraufhin für Verknüpfungen mit Excel-Daten aus Access entfernen.

Unter bestimmten Umständen lässt sich diese Funktionalität jedoch über kleine Umwege wieder reaktivieren: Wenn Sie alleine mit der Excel-Arbeitsmappe arbeiten und während der Arbeit in Access kein anderer Zugriff auf die Arbeitsmappe notwendig ist und wenn im betreffenden Tabellenblatt keine Formeln oder andere individuelle Erweiterungen verwendet werden, lässt sich das folgende Konzept nutzen: Beim Öffnen der Datenbank wird über das Makro "AutoExec" eine VBA-Funktion aufgerufen, die den Inhalt des Tabellenblattes in eine Access-Tabelle importiert. Mit den Inhalten dieser Tabelle können Sie dann in Access arbeiten, also Daten ändern, löschen oder hinzufügen. Beim Schließen der Datenbank wird die Tabelle wieder zurück in die Excel-Arbeitsmappe exportiert, so dass dort die in Access bearbeiteten, aktuellen Daten wieder zur Verfügung stehen.

Sehen wir uns zunächst einmal an, wie der Import der Excel-Daten beim Öffnen der Datenbank automatisiert wird:

1. Löschen Sie zunächst die Verknüpfung mit der Excel-Tabelle im Bereich "Tabellen" des Datenbankfensters bzw. des Navigationsbereichs.

2. Wechseln Sie in die VBA-Entwicklungsumgebung (Alt+F11), legen Sie ein neues Modul "modExcelImpExp" an und kopieren Sie die folgende Funktion "ImportExcelTab()" in dieses Modul:

Function ImportExcelTab(strWorkbook As String, _
                        strExcelTabName As String, _
                        strAccTabName As String)

  On Error Resume Next
  ImportExcelTab = False 'Vorgabe: Problem
  DoCmd.SetWarnings False
  DoCmd.RunSQL "DROP TABLE " & strAccTabName
  DoCmd.SetWarnings True
  Err = 0
  DoCmd.TransferSpreadsheet acImport, _
        acSpreadsheetTypeExcel8, _
        strAccTabName, _
        strWorkbook, _
        True, _
        strExcelTabName & "!"
  If Err <> 0 Then
    Beep
    MsgBox "Fehler beim Importieren von '" & _
           strWorkbook & _
           "' (Tabelle: " & strExcelTabName & ")" & _
           vbCrLf & vbCrLf & Err.Description, _
           vbOKOnly + vbExclamation, _
           "!!! Problem !!!"
  Else
    ImportExcelTab = True
  End If

End Function
· Speichern Sie die Änderungen und verlassen Sie die VBA-Entwicklungsumgebung wieder.

· Legen Sie ein neues Makro "AutoExec" an oder ergänzen Sie ein vorhandenes Makro "AutoExec" am Anfang um die folgende Aktion:

Aktion

AusführenCode

Funktionsname

ImportExcelTab("<Arbeitsmappe>",
               "<ExcelTabelle>",
               "<AccessTabelle>")

Für "Arbeitsmappe" geben Sie Pfad und Dateiname der Excel-Arbeitsmappe an, aus der ein Tabellenblatt importiert werden soll. "ExcelTabelle" definiert das zu importierende Tabellenblatt und "AccessTabelle" ist der Name der Tabelle, die die importierten Daten speichert. Diese Tabelle muss nicht vorhanden sein. Die Angaben müssen in einer Zeile eingetippt werden, wir haben sie hier nur der Übersichtlichkeit halber über mehrere Zeilen verteilt!

· Speichern Sie das Makro und verlassen Sie den Makroentwurf.

Damit verfügen Sie nun über einen Mechanismus, der eine Excel-Tabelle beim Öffnen der Datenbank automatisch importiert.

Die Funktion "ImportExcelTab()" arbeitet folgendermaßen: Im ersten Schritt wird die Tabelle gelöscht, in die die Daten aus der Excel-Tabelle importiert werden sollen. Andernfalls würde die folgende Ausführung von "DoCmd.TransferSpreadsheet" scheitern, da es keine vorhandenen Tabellen ersetzen oder überschreiben kann. Das Löschen der Tabelle übernimmt eine SQL-Anweisung, die per "DoCmd.RunSQL" nicht nur schneller ausgeführt wird, sondern wesentlich einfacher ist, als das übliche Löschen über die "TableDefs"-Auflistung. Per "DoCmd.SetWarnings" schalten wir zuvor die Sicherheitsabfragen von Access aus und anschließend wieder ein. Der zweite Schritt besteht darin, per "DoCmd.TransferSpreadsheet" aus der jeweiligen Arbeitsmappe (Parameter "strWorkbook" die gewünschte Excel-Tabelle (Parameter "strExcelTabName") in die Access-Tabelle (Parameter "strAccTabName") zu übertragen. Wichtig dabei ist, dass der Name der Excel-Tabelle um ein Ausrufezeichen ergänzt wird, damit Excel erkennt, dass ein komplettes Tabellenblatt adressiert wird. Kommt es beim Import zu einem Fehler, wird er entsprechend angezeigt.

Den Weg zurück, also den Export in die Excel-Arbeitsmappe, erläutern wir im zweiten Teil dieses Tipps der Woche in der nächsten Ausgabe von Access-Weekly.

Im ersten Teil dieses Tipps der Woche hatten wir erläutert, wie Sie den Inhalt eines Excel-Tabellenblattes automatisch beim Öffnen der Datenbank in eine Access-Tabelle importieren. In dieser und der nächsten Woche erläutern wir nun den Weg zurück, also den automatischen Export der Access-Tabelle beim Schließen der Datenbank. Dazu ist zunächst das im ersten Teil angelegte Modul "modExcelImpExp" um eine Referenz und eine Funktion zu erweitern:

· Wechseln Sie mit Alt+F11 in die VBA-Entwicklungsumgebung und setzen Sie über Extras-Verweise eine Referenz auf "Microsoft DAO Object Library 3.6" sowie auf "Microsoft Excel x.0 Object Library" (x.0= 12.0 für Excel 2007, 11.0 für Excel 2003, 10.0 für Excel 2002/XP oder 9.0 für Excel 2000).

· Öffnen Sie das Modul "modExcelImpExp".

· Kopieren Sie die folgende Funktion "ExportExcelTab()" in das Modul:

Function ExportExcelTab(strWorkbook As String, _
                        strExcelTabName As String, _
                        strAccTabName As String)

  Dim objExcel As Excel.Application
  Dim wb As Excel.Workbook
  Dim ws As Excel.Worksheet
  Dim strNewName As String

  ExportExcelTab = False 'Vorgabe: Problem
  On Error GoTo ExcelError
  Set objExcel = CreateObject("Excel.Application")
  Set wb = objExcel.Workbooks.Open(strWorkbook)
  On Error Resume Next
  Err = 0
  objExcel.Visible = True
  Set ws = wb.Sheets(strExcelTabName)
  If Err = 0 Then
   'Tabellenblatt vorhanden, umbenennen mit DZ-Stempel
    strNewName = ws.Name & " " & _
                 Format$(Now, "yyyy-mm-dd hh-nn")
    ws.Name = strNewName
    ws.Move After:=wb.Sheets(wb.Sheets.Count)
    Set ws = wb.Sheets.Add
    ws.Name = strExcelTabName
    ws.Move Before:=wb.Sheets(1)
    ws.Range("A1").Select
  End If
  wb.Close True

  Err = 0
  DoCmd.TransferSpreadsheet acExport, _
        acSpreadsheetTypeExcel8, _
        strAccTabName, _
        strWorkbook, _
        True, _
        strExcelTabName & "!"
  If Err <> 0 Then
    Beep
    MsgBox "Fehler beim Exportieren von Tabelle '" & _
           strAccTabName & _
           "' in '" & strWorkbook & "'" & _
           " (Tabelle: " & strExcelTabName & ")" & _
           vbCrLf & vbCrLf & Err.Description, _
           vbOKOnly + vbExclamation, _
           "!!! Problem !!!"
  Else
    ExportExcelTab = True
  End If

ExitFunc:
  On Error Resume Next
  objExcel.Quit
  Set ws = Nothing
  Set wb = Nothing
  Set objExcel = Nothing
  DoEvents
  Err.Clear
  Exit Function

ExcelError:
  Beep
  MsgBox "Fehler beim Zugriff auf Excel: " & _
         Err.Description, _
         vbOKOnly + vbExclamation, _
         "!!! Problem !!!"
  Resume ExitFunc

End Function
· Speichern Sie die Änderungen und verlassen Sie die VBA-Entwicklungsumgebung wieder.

Damit sind die Voraussetzungen für den Export in eine Excel-Arbeitsmappe geschaffen. Wie der Aufruf dieser Funktion automatisch beim Schließen der Datenbank erfolgt, erläutern wir im letzten Teil.

Beschäftigen wir uns an dieser Stelle zunächst damit, wie die Funktion "ExportExcelTab()" arbeitet: Um sicherzustellen, dass Daten in der Excel-Arbeitsmappe nicht unwiederbringlich überschrieben werden, legen wir vor dem Export grundsätzlich eine Sicherungskopie des betreffenden Tabellenblattes unter dem ursprünglichen Namen, ergänzt um einen Datums-/Zeitstempel, an. Dazu wird die Excel-Arbeitsmappe per Office Automation geöffnet, das Tabellenblatt über die "Sheets"-Auflistung adressiert und wie erläutert umbenannt. Danach verschieben wir es mit der Methode "Move" an das Ende der Tabellenblätter. Anschließend legen wir ein neues, leeres Tabellenblatt an, benennen es wie das Tabellenblatt, in das die Daten geschrieben werden sollen, verschieben es an den Anfang der Tabellenblätter und aktivieren die erste Zelle. Dann schließen wir die Arbeitsmappe wieder, damit es keine Konflikte mit dem gleich folgenden "DoCmd.TransferSpreadsheet" gibt. Der Parameter "True" sorgt dafür, das die Arbeitsmappe ohne Rückfrage "Änderungen speichern?" gesichert wird. Der zweite Schritt besteht nun darin, per "DoCmd.TransferSpreadsheet" die Access-Tabelle (Parameter "strAccTabName") in die jeweilige Arbeitsmappe (Parameter "strWorkbook" und dort in die gewünschte Excel-Tabelle (Parameter "strExcelTabName") zu übertragen. Wichtig ist auch hier wieder, dass der Name der Excel-Tabelle um ein Ausrufezeichen ergänzt und so ein komplett adressiertes Tabellenblatt kenntlich gemacht wird. Kommt es beim Export zu einem Fehler, wird er entsprechend angezeigt.

Damit haben wir die Grundlagen, für den Export der Access-Tabelle in eine Excel-Arbeitsmappe abgeschlossen. Wie das automatisch beim Schließen einer Datenbank erfolgt, erläutert der dritte und letzte Teil dieses Tipps der Woche in der nächsten Ausgabe.

In den ersten beiden Teilen dieses Tipps der Woche hatten wir die Basis für den automatisierten Import sowie für den Export von Excel-Daten geschaffen. Bleibt noch, den Export beim Schließen einer Datenbank zu automatisieren: Access stellt mit dem "AutoExec"-Makro zwar eine Möglichkeit bereit, Aktionen beim Öffnen einer Datenbank automatisch auszuführen, aber ein "AutoClose"-Makro oder einen vergleichbaren Mechanismus, der beim Schließen Aktionen automatisch ausführt, gibt es nicht. Für den Export der geänderten Excel-Daten müssen wir also zu einem Trick greifen: Wir setzen ein Formular ein, das beim Öffnen der Datenbank automatisch versteckt über "AutoExec" geladen wird. Beim Verlassen der Datenbank schließt Access der Reihe nach alle offenen Formulare und ruft dabei die Ereignisprozedur "Beim Schließen" des jeweiligen Formulars und somit auch die Ereignisprozedur des unsichtbaren Formulars auf. Hier werden dann die für den Export notwendigen Anweisungen ausgeführt. Das könnte dann zum Beispiel so aussehen:

Private Sub Form_Close()
  Dim fOK As Boolean

  fOK = ExportExcelTab("Z:\Test\Testmappe.xls", _
                       "TabelleKunden", _
                       "ExcelImport")

End Sub

Die zu exportierenden Daten sollen im Tabellenblatt "TabelleKunden" einer Arbeitsmappe "Testmappe.xls" im Verzeichnis "Z:\Test" abgelegt werden. Der Ursprung der Daten ist die Access-Tabelle "ExcelImport", die von der Funktion "ImportExcelTab()" beim Öffnen der Datenbank automatisch angelegt/gefüllt wurde. Diese Parameter übergeben wir dann an die Funktion "ExportExcelTab()", die im Modul "modExcelImpExp" enthalten ist und die wir in der letzten Woche implementiert hatten.

Um die Lösung nun um einen automatischen Export beim Schließen der Datenbank zu komplettieren, gehen Sie wie folgt vor:

1. Legen Sie ein Formular "frmAutoClose" an.

2. Lassen Sie die Ereignisprozedur "Beim Schließen" des Formulars anzeigen und kopieren Sie folgende Deklaration und Anweisung in den VBA-Editor:

  Dim fOK As Boolean

  fOK = ExportExcelTab("Z:\Test\Testmappe.xls", _
                       "TabelleKunden", _
                       "ExcelImport")
· Passen Sie Pfad/Dateiname der Arbeitsmappe, Name des Tabellenblattes und Name der Access-Tabelle entsprechend an.

· Fügen Sie im Makro "AutoExec" am Anfang eine Aktion "ÖffnenFormular" hinzu, wählen Sie im Feld "Formularname" den Eintrag "frmAutoClose" und im Feld "Fenstermodus" den Eintrag "Ausgeblendet" aus.

· Sichern Sie die Änderungen am Makro und verlassen Sie den Makroentwurf.

Beim Öffnen der Datenbank wird "AutoExec" automatisch ausgeführt und darüber sichergestellt, dass das Formular "frmAutoClose" versteckt angezeigt wird, damit über dessen Ereignisprozedur "Beim Schließen" später der Export erfolgen kann. Im Anschluss wird die Excel-Tabelle importiert und Sie können wie gewohnt damit arbeiten. Beim Schließen der Datenbank werden dann die Daten über die Ereignisprozedur "Beim Schließen" wieder zurück in die Excel-Arbeitsmappe transportiert.
